Séance n°2
Procédures usuelles - Graphiques

Un programme SAS est composé d’une ou plusieurs étapes DATA (cf. séance n°1) suivies d’une ou plusieurs étapes PROC (ou procédures). Une étape PROC applique des traitements à des tables et génère des sorties, des graphiques ou encore de nouvelles tables. Les procédures PRINT, SORT, IMPORT et EXPORT ont déjà été abordées au cours de la séance n°1. Pour toutes informations supplémentaires se référer à l’aide de SAS en ligne à l’adresse suivante :   http://support.sas.com/onlinedoc/913/docMainpage.jsp
Rappel : Licence SAS gratuite à domicile
Pour disposer d’une version de SAS utilisable chez vous uniquement, vous devez vous rendre à cette adresse : http://www.sas.com/offices/europe/france/academic/licence-gratuite-SAS-domicile.html.

Remplissez le formulaire en ligne, imprimez-le, datez et signez, et ensuite donnez-le à votre délégué qui le remettra à l’enseignant qui le remettra au CRI diffuseur du logiciel.

1. Compléments sur le TP1

Attention, sous Linux, un chemin de fichier s’écrit de la façon suivante : 
"/home/11111/ TPSAS /exemple.txt" où /home/11111/ est votre « home », auquel on peut accéder en utilisant le raccourci ~ (ex : « ~/tpsas/exemple.txt »). Le code 11111 représente votre login.
La ligne de commande LINUX pour copier le dossier datasas est la suivante :

cp -R  /home/saintpierreg/TPSAS/datasas/ /home/11111/TPSAS/datasas/

Les TP seront ranges dans des dossiers TP1, TP2, etc. rangés dans le dossier TPSAS.

Exporter un tableau de données dans un fichier externe :

Une étape data peut aussi servir à créer un fichier externe avec les données d’une table SAS existante. Cette étape DATA diffère de celles vues précédemment puisque l’objectif n’est pas ici de construire une nouvelle table mais d’utiliser les éléments d’une table SAS pour générer un fichier texte « .txt » ; on utilisera alors une instruction DATA _NULL_;. L’instruction FILE indique ensuite le chemin avec le nom du fichier qui doit recevoir les données exportées. L’instruction PUT s’utilise de manière analogue à l’instruction INPUT pour indiquer le nom des variables à considérer. Elle permet, si on le souhaite, d’écrire les données en utilisant un pointeur d’écriture (position relative ou absolue pour l’écriture d’une valeur sur la ligne d’un individu) ainsi qu’un format d’écriture (la forme sous laquelle on va écrire la valeur). Ce format pourra aussi bien être un format SAS prédéfini qu’un format utilisateur préalablement défini dans PROC FORMAT (cf. exercice 2, séance n°2). La valeur exportée sera alors la valeur formatée au lieu de celle stockée suivant le format par défaut de SAS. Par rapport à INPUT, il n’est pas nécessaire de signaler les variables nominales par un « $ » ; lors de l’export, le type des variables SAS est déjà connu.
Syntaxe pour un fichier « .txt » :
filename file3 '~ / … / fichier.txt';

DATA _NULL_;
SET < nom de la Table SAS >;
FILE '~ / … / fichier.txt'
dlm = 'délimiteur';
/* ou FILE file3 avec filename */

PUT < pos_point_1 > < var_1 > : < format_1 > 

… < pos_point_p > < var_p > : < format_p >;

RUN;
Où filename comme dans le TP1 crée le raccourci nommé file3 du fichier dont le nom est fichier.txt. 
Exemple d’export de données

A partir du fichier de données « exemple » utilisé dans le TP1 pour illustrer l’instruction cards, utiliser les instructions suivantes pour l’exporter sous forme texte.
DATA _NULL_;

SET exemple;

FILE '~/TPSAS/datasas/exemple.txt';

PUT nom 1-6
@9 naiss:DATE9.
+1 sin98;

RUN;
Dans l’exemple, les données relatives à la variable nom sont affichées sur les colonnes 1 à 6 de la ligne d’un individu (tant pis si le nom a plus de 6 caractères !). Le curseur se place ensuite en position 9 (soit en 8ème colonne) pour afficher la date de naissance naiss dans le format prédéfini DATE.9. Enfin, lorsqu’une valeur est inscrite, le curseur laisse (par défaut) un espace pour inscrire la valeur suivante ; avec la commande +1, le curseur est déplacé d’une colonne supplémentaire pour finalement afficher les données de sin98.
Vérifier dans le dossier '~/TPSAS/datasas/’ que le fichier exemple.txt s’y trouve.
Pour exporter des données dans un fichier externe, nous pourrons encore utiliser la commande File > Export Data (plus simple) ou bien une PROC EXPORT (plus compliqué).

PROC EXPORT DATA = exemple
           OUTFILE= "~/TPSAS/datasas/exemple.txt"
           DBMS=DLM REPLACE;

     DELIMITER='20'x; /* delimiter = espace */
RUN;
Bien sur, d’autres délimiteurs peuvent être utilisés (DELIMITER='&' par exemple). Par défaut le délimiteur est un espace, et il n’y a pas besoin d’en spécifier un si le nom du fichier crée est un .txt ou un .xls (rechercher « procedure export » dans l’aide pour plus de précisions).

2. Procédures usuelles

Syntaxe

Le début de l’étape est signalé par PROC suivi du nom de la procédure, puis du nom de la table à traiter. Il est important de spécifier la table sur laquelle on travaille avec l’instruction « DATA = < nom_tab > ». Par défaut, la procédure agit sur la dernière table créée. La fin de l’étape est reconnaissable à l’instruction RUN.

PROC < nom de la procédure > DATA = < nom de la Table SAS > < options >;

...
RUN;
Exemple : impression de la table exemple.
libname malib '~ / … /TPSAS/datasas';

proc print data=malib.exemple; run;

Exemples d’instructions usuelles pour quelques procédures de SAS/BASE:

Nous vous invitons fortement à étudier la syntaxe précise des procédures que vous rencontrerez à partir de l’aide en ligne de sas qui se trouve à l’adresse suivante :

http://support.sas.com/onlinedoc/913/docMainpage.jsp
· L’instruction PROC (abréviation de procédure) est utilisée pour appeler une procédure SAS.


PROC < nom_proc > DATA = < nom_tab > < options >;
· L’instruction VAR (abréviation de variables) est suivie de la liste des variables à considérer.


VAR < liste des variables >;
· L’instruction BY permet le traitement par sous-groupes.


BY < sous-groupe de variables >;
Remarque
La table SAS traitée par l’instruction BY doit avoir été triée suivant les variables mentionnées dans l’instruction.

· L’instruction TITLE définit le texte à imprimer en en-tête des pages de sortie. On peut ajouter jusqu’à 10 lignes de titres.


TITLEn 'titre';
où le n spécifie le nième titre.

· L’instruction FOOTNOTE définit le texte à imprimer en bas des pages de sortie. On peut ajouter jusqu’à 10 lignes de bas de pages.


FOOTNOTEn 'bas de page';
où le n spécifie le nième bas de page.

Titres SAS : TITLE et FOOTNOTE sont des instructions globales ; elles peuvent être utilisées n’importe où dans le programme (étape DATA, étape PROC ou en dehors). Quand on modifie un titre, tous les titres hiérarchiquement inférieurs sont effacés. Pour supprimer les titres, on pourra utiliser l’instruction TITLE sans argument TITLE;.

Quelques exemples de procédures usuelles (liste non exhaustive…)

· Certaines de ces procédures agissent directement sur des tables SAS :

PROC PRINT : procédure d’impression ;

PROC SORT : procédure de tri ;

PROC STANDARD pour centrer et réduire les données d’une table ;

PROC STANDARD
DATA=nom_tab
VARDEF=n
MEAN=0
STD=1
OUT=reduit;

RUN;
/* VARDEF précise le diviseur dans le calcul de la variance */

PROC TRANSPOSE pour échanger les variables et les observations d’une table et réciproquement ;

· D’autres calculent des statistiques simples :

PROC MEANS édite des statistiques simples : moyenne, écart-type, maximum et minimum sur les variables numériques uniquement. Des options permettent d’obtenir des statistiques supplémentaires : SUM, SKEWNESS (coefficient d’asymétrie), KURTOSIS (coefficient d’aplatissement), T pour un test t de Student sur la moyenne… Il est possible de restreindre les variables d’étude avec l’instruction VAR ;

PROC MEANS
DATA = nom_tab
< options means >;

RUN;

PROC CORR pour calculer les corrélations entre toutes les variables numériques ;

PROC UNIVARIATE produit certaines statistiques univariées pour chaque variable numérique considérée ;

PROC FREQ pour calculer les effectifs par modalité d’une variable discrète et des tests du Chi2 ;

· D’autres permettent de construire des modèles :

PROC REG pour effectuer un modèle de régression linéaire reliant une (ou plusieurs) variables(s) expliquée(s) (ou dépendante(s)) avec une (ou plusieurs) variable(s) descriptive(s) (ou prédicteur(s)) ;

PROC REG
DATA=tabin
< options reg >;

MODEL
< var_dép > = < var_exp > / < options >;

PLOT
< var_dép > * < var_exp > = 'symbol';

OUTPUT OUT=tabout
P=< nom1 >
R=< nom2 >;

/*
sauvegarde des prédicteurs et des résidus dans une table tabout
*/
RUN;

PROC GLM pour effectuer des modèles d’analyse de la variance non-orthogonaux, d’analyse de la covariance et des modèles linéaires généraux ;

PROC GLM
DATA=tabin
< options glm >;

CLASS
< variable(s) >;

MODEL
< var_dép > = < var_exp >;

OUTPUT OUT=tabout
keyword = < nom_de_variable >;

RUN;

PROC GENMOD pour effectuer un modèle de régression linéaire généralisée. La régression logistique (disponible aussi avec la PROC LOGISTIC) est notamment une régression linéaire généralisée particulière lorsque la fonction de lien (« link function ») est la fonction LOGIT.

Faire les exercices 1 et 2.
3. Graphiques à 2 dimensions

· NUAGES DE POINTS

PROC PLOT produit des graphiques « en basse résolution » (avec des caractères) de nuages de points (scatter plots) à 2 dimensions ;

PROC PLOT
DATA = nom_tab
< options plot >;

PLOT < Y_var > * < X_var > = 'caractère' / < plot options >;

RUN;

Les options PLOT sont nombreuses, elles permettent entre autres de préciser les axes, les marques d’échelles et leurs libellés, de tracer des lignes supplémentaires, d’encadrer le graphe (BOX) ou de superposer plusieurs graphes (OVERLAY). Quelques options PLOT : BOX, OVERLAY, VREF = , HREF = …

Remarque
En exécutant la proc PLOT sur un exemple, on pourra apprécier les faiblesses des graphiques affichés dans la fenêtre Output ! On préférera donc utiliser la PROC GPLOT qui permet d’obtenir des graphiques « en haute résolution » et dont la syntaxe est explicitée dans l’exercice n°4. Même remarque pour la proc CHART décrite ci-dessous, on utilisera plutôt la PROC GCHART.

· DIAGRAMMES

PROC CHART trace « en basse résolution » des histogrammes horizontaux (hbar), verticaux (vbar), bivariés (block), des camemberts (pie) et des diagrammes en étoiles (star). Elle traite des variables quantitatives ou qualitatives. Les variables quantitatives sont codées explicitement ou automatiquement en classes.

PROC CHART
DATA = nom_tab
< options chart >;

VBAR < variable > / < chart options >;
( édite des histogrammes verticaux

HBAR < variable > / < chart options >;
( édite des histogrammes horizontaux

PIE < variables > / < chart options >;
( édite des camemberts…

RUN;

Exemples d’options CHART :

· SUMVAR = 
variable numérique dont le cumul ou la moyenne est représenté ;

· DISCRETE
spécifie que les valeurs sont de type discret (avec un nombre limité de modalités) ;

· MIDPOINTS = < valeurs >
spécifie le point moyen de chaque bâton d’un histogramme, par exemple

MIDPOINTS = 10 TO 100 BY 10;
( définit un histogramme à 10 bâtons centrés sur les valeurs 10, 20, …, 100 ;

MIDPOINTS = 'bleu' 'vert' 'rouge';
( définit un histogramme à 3 bâtons pour les valeurs discrètes citées ;

· GROUP = < variable >
représentation de plusieurs graphes côte à côte suivant les modalités d’une même variable ;

· SUBGROUP = < variable >
découpage des barres ou colonnes selon la contribution de chacune des  modalités d’une variable ;

· LEVELS = < nombre de classes > ;

· TYPE = 
spécifie ce que représente le graphique (par défaut une fréquence FREQ) ; autrement on peut choisir : CFREQ (fréquence cumulée), CPT (pourcentage cumulée), PCT (pourcentage)…

Dans les exercices qui suivent les tables utilisées se trouvent dans le répertoire datasas. Pensez à bien spécifier les chemins vers les fichiers concernés.
Exercice n°1

La table note est extraite de « Multivariate Analysis », K.V. Mardia, J.T. Kent et J.M. Bibby. Ces notes sont relatives à 88 étudiants d’une université anglaise ayant passé 5 épreuves notées par des notes entières de 0 à 100. Les 5 épreuves sont respectivement : mécanique, algèbre linéaire, algèbre des structures, analyse et statistique. En plus des 5 colonnes correspondant aux 5 épreuves, une première colonne indique le numéro de l’étudiant (de 01 à 88). Le but de cet exercice est d’observer les effets des lignes de code ci-dessous. 
options pagesize=64 linesize=78 nodate; title; footnote 'TP2: Exercice 1';

libname malib '~ / … /Cours de SAS/TPSAS/datasas';

/* Sample Data Set: note */

proc print data=malib.note; run;

proc means data=malib.note /* vardef=n */; run;

proc univariate data=malib.note /* plot vardef=n */; run;

proc corr data=malib.note < options >; run;

Commentaires
Par défaut, la proc CORR calcule le coefficient de corrélation de Pearson. Autrement, on pourra spécifier l’option

· SPEARMAN : coefficient de corrélation des rangs de Spearman ; ou KENDALL : coefficient de corrélation de Kendall ; ou HOEFFDING.

D’autres options sont aussi disponibles parmi lesquelles :

· NOSIMPLE : pour supprimer l’impression des statistiques simples ;

· COV : pour imprimer les covariances (option incompatible avec les options SPEARMAN et KENDALL !) ;

· OUTP = nom_tab : pour créer une table SAS contenant les coefficients de corrélation de Pearson.

proc plot data=malib.note;

plot stat*meca;

plot anls*algb;

run;

proc standard data=malib.note vardef=n mean=0 std=1 out=reduit; run;

proc print data=reduit; run;

proc corr data=reduit cov vardef=n; run;
/* expliquer l’effet de l’option vardef dans les proc standard et corr*/
Exercice n°2

options pagesize=28 nodate; title; footnote 'TP2: Exercice 2';

libname malib '~/TPSAS/datasas ';

/* Sample Data Set: htwt */

/* Décrire la table de données et noter l'effet de la commande "(obs=10)" */

proc print data=malib.htwt (obs=10); run;

proc sort data=malib.htwt out=TRI1;

by name;

run;

proc print data=TRI1 (obs=10);

id name;

var sex age;

title 'PROC SORT: Exemple 1';

run;

proc sort data=malib.htwt out=TRI2;

by sex age;

run;

proc print data=TRI2 (obs=10);

id name;

var sex age;

title 'PROC SORT: Exemple 2';

run;

proc means data=malib.htwt;

title 'PROC MEANS: Exemple 1';

run;

/* Notez les effets des options suivantes:

n, mean, var, min, max, sum, nmiss et maxdec=1 */

proc means data=malib.htwt n mean var min max sum nmiss maxdec=1;

var age;

class sex;

title 'PROC MEANS: Exemple 2';

run;

proc corr data=malib.htwt;

var age height weight;

title 'PROC CORR Exemple';

run;

proc univariate data=malib.htwt;

var age;

title1 'PROC UNIVARIATE Exemple';

run;

proc format;

value $sexl

'M' = 'Male'
'F' = 'Female';

value agef

low-29 = '1'
30-39 = '2'
40-49 = '3'
50-high = '4';

run;

proc print data=htwt;

format sex $sexl. age agef.;

run;
Exercice n°3 : comment simuler des observations suivant une loi donnée ?

/* On prendra par exemple: N = 1000 et s = 10. */

title 'Simulation de 1000 observations d'une loi gaussienne'; footnote 'TP2 Exercice3';

data simul;
do i=1 to N;
( pour un échantillon de N observations

y=rannor(s);
( pour une distribution Gaussienne centrée réduite

output;

end;

keep y;

run;

On pourra simuler d’autres lois en utilisant les instructions : ranbin (binomiale), rancau (Cauchy), ranexp (exponentielle), rangam (gamma), ranpoi (Poisson), ranuni (uniforme [0,1]) etc. Les tirages sont des tirages pseudo-aléatoires. Entre deux procédures de tirage, il importe de changer la valeur de la racine d’initialisation s (seed). Autrement, les tirages sont tous identiques ! Par défaut, SAS choisit s=0.

Utilisez l’aide en ligne de sas pour voir la syntaxe appropriée pour chaque fonction.

proc gchart data=simul;

title 'Histogramme des observations';

vbar y; /* vbar3d y; */
run;

quit;

Exercice n°4 : exemple d’utilisation de la proc GPLOT

title; footnote 'TP2: Exercice4';

libname malib '~/TPSAS/datasas';

/***************************************************/
/*
Sample Data Set: tabac
*/
/*
Sources: Direction des Services Fiscaux, ISEE
*/
/***************************************************/
goptions reset=all rotate=landscape border ftext=hwdmx013 htext=0.25 cm;

title1 c=blue f=hwdmx020 h=1.4 j=center 'Evolution de la consommation de cigarettes francaises et etrangeres entre 1989 et 2004';

title2 c=bl f=hwdmx011 h=1 'Source: ISEE';

symbol1 value=dot interpol=join line=1 color=blue width=2;

symbol2 v=circle i=join l=8 c=green w=2;

symbol3 v=square i=join l=15 c=red w=2;

symbol4 v=diamond i=splines l=22 c=yellow w=2;

symbol5 v=# i=needle l=29 c=black w=2;

symbol6 v=none i=step l=34 c=purple w=2;

axis1
length=17 cm


label=(h=1 f=hwdmx021 'Annees')


value=(f=hwdmx021 h=0.8)


order=(1989 to 2004 by 1)


minor=(h=0.2 width=1.0 n=4);

axis2
length=10.6 cm


label=(h=1 f=hwdmx021 'Consommation en kg')


value=(f=hwdmx021 h=0.8)


order=(500 to 190500 by 10000)


major=(h=0.5 width=1.5)


minor=(h=0.2 width=1.0 n=4);

legend1 frame cframe=ligr position=(bottom center outside);

proc gplot data=malib.tabac gout=graf1;

plot cigfr*annee=1 ciget*annee=2 roule*annee=3
/ overlay haxis=axis1 vaxis=axis2 legend=legend1;

run;

quit;

Exercice n°5 : exemple du cours, tableau de contingence et proc FREQ

Data CEREQ;

input Diplome $ 1-8 Situ $ 11-17 Poids;

put _infile_;

cards;

NON DIPL  CHOMAGE 54

NON DIPL  MESURE 52

NON DIPL  EMPLOI 40

DIPLOMES  CHOMAGE 122

DIPLOMES  MESURE 97

DIPLOMES  EMPLOI 133

;

run;

Y a t'il un lien entre le diplôme et l'insertion des jeunes sur le marché du travail ?

Pour répondre il faut calculer le KHI-2 (cf page 205 poly Josiane Confais)

http://www-rocq.inria.fr/axis/modulad/archives/numero-33/tutorial-confais-33/confais-33-tutorial.pdf
Proc FREQ data=cereq order=data;

tables Diplome*Situ;

weight Poids;

Title 'Tables Diplome*Situ';

Title3 "Source : Enquête d'insertion CEREQ-DEP";

Title4 'de terminale CAP au BEP Commerce en L.P. (SN, Apprentis exclus)';

run;

